Q: How to convert Unicode to Chinese I want
"\u5b66\u6821\u73b0\u6709\u5728\u6821\u5b66\u751f25000\u4f59\u4eba\u\ ff0c\u73b0\u5f00\u8bbe30\u4e2a\u672c\u79d1\u4e13\u4e1a"
to
"学校现有在校学25000生余人,现开30设个本科专业"
I tried this
Unicode[in_String] := Module[
   {CustomaryUnicode = in, Chinese, position, myInsert},
   Chinese = StringCases[CustomaryUnicode, 
     RegularExpression["(?<=\\\u)(.{4})"]];
   position = StringCases[CustomaryUnicode, 
     RegularExpression["((?<=\\\u)([^\\\]{4})([^\\\]+))"] -> {"$2", 
   "$3"}];
   TruePosition = Position[Chinese, #] & /@ Cases[position, {x_, _} -> x] // 
     Flatten[#, 1] &;
   myInsert[l_, e_, p_] := 
    Module[{i = 1}, 
     ReleaseHold@Insert[l, Hold[Sequence @@ e[[i++]]], p]];
   StringJoin@myInsert[
     FromCharacterCode[
      Cases[FromDigits[#, 16] & /@ Chinese, x_ -> {x}],
      "UTF-8"],
     Cases[position, {_, x_} -> {x}],
     TruePosition]
   ]; 


But it's not perfect,For example, in this case
"\u56ed\u6821\u4e8e\u9633\u533a\u5e73\u4e50\u56ed100\u53f7"
If the Unicode in front of the number appears more than once, the return value of the "Position" in the function will become two and I don't know which is correct

A: Mathematica escapes unicode characters as \:262F instead of the usual \u262F (documentation). With help from @rnotlnglgq:
toUnicode = ToExpression@*StringReplace["\u" -> "\:"];

toUnicode @ "\u5b66\u6821\u73b0\u6709\u5728\u6821\u5b66\u751f25000\u4f59\u4eba\uff0c\u73b0\u5f00\u8bbe30\u4e2a\u672c\u79d1\u4e13\u4e1a"

(*    "学校现有在校学生25000余人,现开设30个本科专业"    *)

If the syntax warnings worry you, escape the backslashes properly:
toUnicode = ToExpression@*StringReplace["\\u" -> "\\:"];

toUnicode @ "\\u5b66\\u6821\\u73b0\\u6709\\u5728\\u6821\\u5b66\\u751f25000\\u4f59\\u4eba\\uff0c\\u73b0\\u5f00\\u8bbe30\\u4e2a\\u672c\\u79d1\\u4e13\\u4e1a"
(*    "学校现有在校学生25000余人,现开设30个本科专业"    *)


A: The rule of replacing can be:
StringReplace["\\u"~~s:Repeated[HexadecimalCharacter,{4}]:>FromCharacterCode@FromDigits[s,16]]

BTW, if you are handling JSON, the following direct way works:
ImportString["\"\\u79d1\"","JSON"]

